Introduction to SML
Finite Trees

Michael R. Hansen

mrh@imm.dtu.dk

Informatics and Mathematical Modelling

Technical University of Denmark

Overview

Finite Trees

- Algebraic Datatypes.
- Recursions following the structure of trees.
- Abstract datatypes.
- Examples
  - Search trees
  - Expression trees
  - Regular Expressions

Trees

A *finite tree* is a value which may contain a subcomponent of the same type.

Example: A *binary search tree*



Condition: for every node containing the value x: every value in the left subtree is smaller then x, and every value in the right subtree is greater than x.

Binary Trees

A *recursive datatype* is used to represent values with are trees.

```
datatype tree = Lf | Br of tree*int*tree;
> datatype tree
> con Lf = Lf : tree
> con Br = fn : tree * int * tree -> tree
```

Binary Trees

A *recursive datatype* is used to represent values with are trees.

```
datatype tree = Lf | Br of tree*int*tree;
> datatype tree
> con Lf = Lf : tree
> con Br = fn : tree * int * tree -> tree
```

The two parts in the declaration are rules for generating trees:

- `Lf` is a tree
- if t_1, t_2 are trees, n is an integer, then `Br`(t_1, n, t_2) is a tree.

Binary Trees

A *recursive datatype* is used to represent values with are trees.

```
datatype tree = Lf | Br of tree*int*tree;
> datatype tree
> con Lf = Lf : tree
> con Br = fn : tree * int * tree -> tree
```

The two parts in the declaration are rules for generating trees:

- `Lf` is a tree
- if t_1, t_2 are trees, n is an integer, then `Br`(t_1, n, t_2) is a tree.

The tree from the previous slide is denoted by:

```
Br(Br(Br(Lf,2,Lf),7,Lf),
   9,
   Br(Br(Lf,13,Lf),21,Br(Lf,25,Lf)))
```

Binary search trees: Insertion

Recursion on the structure of trees:

- Constructors `Lf` and `Br` are used in patterns

```
fun insert(i, Lf)                      = Br(Lf,i,Lf)
  | insert(i, tr as Br(t1,j,t2)) =
        case Int.compare(i,j) of
              EQUAL   => tr
            | LESS    => Br(insert(i,t1),j,t2)
            | GREATER => Br(t1,j,insert(i,t2))
```

- The search tree condition is an invariant for `insert`

Example:

```
- val t1 = Br(Lf, 3, Br(Lf, 5, Lf));

- val t2 = insert(4, t1);
> val t2 = Br(Lf, 3, Br(Br(Lf, 4, Lf), 5, Lf)) : tree
```

Binary search trees: member and toList

```
fun member(i, Lf)           = false
  | member(i, Br(t1,j,t2)) =
        case Int.compare(i,j) of
              EQUAL   => true
            | LESS    => member(i,t1)
            | GREATER => member(i,t2)
> val member = fn : int * tree -> bool
```

In-order traversal

```
fun toList Lf             = []
  | toList(Br(t1,j,t2)) = toList t1 @ [j] @ toList t2;
> val toList = fn : tree -> int list
```

gives a sorted list

```
- toList(Br(Br(Lf,1,Lf), 3, Br(Br(Lf,4,Lf), 5, Lf)));
> val it = [1, 3, 4, 5] : int list
```

Deletions in search trees

Delete minimal element in a search tree: `tree -> int * tree`

```
fun delMin(Br(Lf,i,t2)) = (i,t2)
  | delMin(Br(t1,i,t2)) = let val (m,t1') = delMin t1
                          in (m, Br(t1',i,t2)) end
```

Delete element in a search tree: `tree * int -> tree`

```
fun delete(Lf,_)           = Lf
  | delete(Br(t1,i,t2),j) =
      case Int.compare(i,j) of
           LESS    => Br(t1,i,delete(t2,j))
         | GREATER => Br(delete(t1,j),i,t2)
         | EQUAL   =>
            (case (t1,t2) of
                 (Lf,_) => t2
               | (_,Lf) => t1
               |  _     => let val (m,t2') = delMin t2
                           in Br(t1,m,t2') end)
```

Expression Trees

```
infix 6 ++ --;
infix 7 ** //;

datatype fexpr =
    Const of real
  | X
  | ++ of fexpr * fexpr | -- of fexpr * fexpr
  | ** of fexpr * fexpr | // of fexpr * fexpr

> datatype  fexpr
    con ** : fexpr * fexpr -> fexpr
    con ++ : fexpr * fexpr -> fexpr
    con -- : fexpr * fexpr -> fexpr
    con // : fexpr * fexpr -> fexpr
    con X : fexpr
    con Const : real -> fexpr
```

Symbolic Differentiation D: fexpr -> fexpr

```
fun D(Const _)      = Const 0.0
  | D X             = Const 1.0
  | D(fe1 ++ fe2)   = (D fe1) ++ (D fe2)
  | D(fe1 -- fe2)   = (D fe1) -- (D fe2)
  | D(fe1 ** fe2)   = (D fe1) ** fe2 ++ fe1 ** (D fe2)
  | D(fe1 // fe2)   =
      ((D fe1)**fe2 -- fe1 ** (D fe2)) // (fe2**fe2)
```

Example:

```
D (X ** (X **(Const 3.0 ++ X)));
> val it =
    ++(**(Const 1.0, **(X, ++(Const 3.0, X))),
       **(X,
          ++(**(Const 1.0, ++(Const 3.0, X)),
             **(X, ++(Const 0.0, Const 1.0))))) : fexp
```

Expressions: Computation of values

```
comp : fexpr * real -> real

fun comp(Const r,_)     = r
  | comp(X,y)           = y
  | comp(fe1 ++ fe2,y) = comp(fe1,y) + comp(fe2,y)
  | comp(fe1 -- fe2,y) = comp(fe1,y) - comp(fe2,y)
  | comp(fe1 ** fe2,y) = comp(fe1,y) * comp(fe2,y)
  | comp(fe1 // fe2,y) = comp(fe1,y) / comp(fe2,y)
```

Example:

```
comp(X ** (Const 2.0 ++ X), 4.0);
> val it = 24.0 : real
```

Abstract types

- internal representation is hidden from a user

reuseability, modularization, correctness

Invariants may be protected by hiding the representation

Example: Violation of an invariant for search trees:

```
insert(4, Br(Lf, 5, Br(Lf, 2, Lf)));
> val it = Br(Br(Lf, 4, Lf), 5, Br(Lf, 2, Lf)) : tree
```

Solution: Abstract data types
— search trees are only accessable by the following operations:

```
empty  : stree
insert : int * stree -> stree
member : int * stree -> bool
toList : stree -> int list
```

Abstract types in SML: Search trees (1)

```
abstype stree = Lf | Br of stree * int * stree
with
  val empty = Lf

  fun insert(i, Lf)                    = Br(Lf,i,Lf)
    | insert(i, tr as Br(t1,j,t2)) =
          case Int.compare(i,j) of
                .....

  fun member(i, Lf)              = false
    | member(i, Br(t1,j,t2)) = .....


  fun toList Lf                = []
    | toList(Br(t1,j,t2)) = toList t1 @ [j] @ toList t
end;
```

Abstract types in SML: Search trees (2)

The representation of `stree` is *hidden*:

```
>  abstype stree
>  val empty = <stree> : stree
>  val insert = fn : int * stree -> stree
>  val member = fn : int * stree -> bool
>  val toList = fn : stree -> int list
```

Constructors are invisible:

```
Br(Lf, 5, Br(Lf,0,Lf));
> ! Toplevel input:
> ! Br(Lf, 5, Br(Lf,0,Lf));
> ! ^^
> ! Unbound value identifier: Br
```

— the search tree invariant cannot be violated

Abstract types in SML: Search trees (3)

Examples:

```
val st1 = insert(2, empty);
> val st1 = <stree> : stree

val st2 = insert(~3, insert(7, st1));
> val st2 = <stree> : stree

member(4, st2);
> val it = false : bool

member(7, st2);
> val it = true : bool

toList st2;
> val it = [~3, 2, 7] : int list
```

Regular Expressions

```
infix 4 ++
infix 6 ##

datatype reg = Empty                (* Empty set     *)
             | Epsilon              (* Empty string  *)
             | S of string          (* Symbol        *)
             | ++ of reg * reg      (* Union         *)
             | ## of reg * reg      (* Concatenation *)
             | ** of reg;           (* Kleene star   *)

val a = S "a"
val b = S "b"

- a ++ a ## **b ## a;
> val it =
>   ++(S "a", ##(##(S "a", **(S "b")), S "a")) : reg
```

Function on Regular Expressions (1)

```
   regToString = fn : reg -> string

fun regToString Empty       = ""
  | regToString Epsilon     = "<>"
  | regToString (S s)       = s
  | regToString (r1 ++ r2) =
      "(" ^ regToString r1 ^ "+" ^ regToString r2 ^ "
  | regToString (r1 ## r2) =
      "(" ^ regToString r1 ^ "#" ^ regToString r2 ^ "
  | regToString ( ** r )    =
      "(" ^ regToString r ^ ")" ^ "^* " ;

- regToString ( a ++ a ## **b ## a);
> val it = "(a+((a#(b)^* )#a))" : string
```

$w \in L(R)$

```
fun _  match Empty       = false
 | xs match Epsilon     = xs = []
 | xs match (S x)       = xs = [x]
 | xs match (r1 ++ r2) = xs match r1 orelse xs match r
 | xs match (r1 ## r2) =
        matchInRange (0, length xs) xs (r1, r2)
 | [] match ( ** r    ) = true
 | xs match ( ** r    ) =
        matchInRange (1, length xs) xs (r, ** r)

and matchAt i xs (r1,r2) =  List.take(xs,i) match r1
                        andalso List.drop(xs,i) match r2

and matchInRange (i, j) xs (r1, r2) =
    not (i=j+1)
    andalso (matchAt i xs (r1,r2)
             orelse matchInRange (i+1, j) xs (r1, r2))
```

Examples

```
- val r1 = a ++ a ## ** b ## a;

- ["a", "b", "b", "a"] match r1;
> val it = true : bool

-  ["a", "b", "b", "a", "b"] match r1;
> val it = false : bool
```

$L(R) = \{\}$?

```
isE : reg -> bool
```

```
fun isE Empty      = true
  | isE Epsilon    = false
  | isE(S _)       = false
  | isE(r1 ++ r2) = isE r1 andalso isE r2
  | isE(r1 ## r2) = isE r1 orelse isE r2
  | isE( ** r)     = false;
```

- $L(r) = \{\epsilon\}$?
- $L(r)$ is finite?
- $L(r)$ is infinite?

are left as exercises

Example

```
val r1 = ** ( ** Empty ## (Empty ++ ** Empty)) ;
isE r1;
> val it = false : bool

isE(r1##Empty);
> val it = true : bool

isE(r1 ## ** (a ## b ++ Empty));
> val it = false : bool
```